Pensieve header: Hour 30: Perturbing the Heisenberg-algebra knot invariant (2).

Recall,

$$m_k^{ij} = e^{(\xi_i+\xi_j)\,x_k+(\pi_i+\pi_j)\,p_k-\xi_i\,\pi_j},$$

$$R_\epsilon = e^{(T-1)\,(p_i-p_j)\,x_j+R'};\ \ R' = \sum_{k=1}^{\$k} \epsilon^k\, R^{(k)},$$

$${}_A\mathcal{L}_B\ //\ {}_B\mathcal{M}_C = e^{\sum_{i\in B}\partial_{z_i}\partial_{\zeta_i}}(\mathcal{L}\cdot\mathcal{M}),$$

$$\langle F:\mathcal{E}\rangle_B = e^{\frac{1}{2}\sum_{u,v\in B}F_{uv}\,\partial^u\partial^v}\ \mathcal{E}\ |_{z_B=0}\ \text{ and }\ [F:\mathcal{E}]_B = e^{\frac{1}{2}\sum_{u,v\in B}F_{uv}\,\partial^u\partial^v}\ \mathcal{E}$$

(Note, the two are equi-computable: clearly if we know how to compute $[F:\mathcal{E}]$ we also know how to compute $\langle F:\mathcal{E}\rangle$, and also $[F:\mathcal{E}]\ |_{z_B\to\bar{z}_B} = \langle F:\mathcal{E}\ |_{z_B\to z_B+\bar{z}_B}\rangle$).

$Z_\lambda := \log\left[\lambda F : e^E\right]$ satisfies $Z_0 = E$ and the "synthesis equation",

$$\partial_\lambda Z_\lambda = \tfrac{1}{2}\,F_{uv}(\partial_u\,\partial_v\,Z_\lambda + (\partial_u\,Z_\lambda)\,(\partial_v\,Z_\lambda)).$$

Lemma 1. $\left\langle F:\mathcal{E}\,e^{\frac{1}{2}\sum_{i,j\in B}G_{ij}\,z_i\,z_j}\right\rangle_B = \det(I-GF)^{-1/2}\left\langle F(I-GF)^{-1}:\mathcal{E}\right\rangle_B.$

Lemma 2. $\langle F:\mathcal{E}\,e^{\sum_{i\in B}y_i\,z_i}\rangle_B = e^{\frac{1}{2}\sum_{i,j\in B}F_{ij}\,z_i\,z_j}\left\langle F:\mathcal{E}\ |_{Z_B\to Z_B+Fy_B}\right\rangle_B.$

To solve the synthesis equation in general, we write $Z_\lambda = \sum Z[m]\,\lambda^m$ and then solve iteratively $Z[0] = Z_0 = E$ and

$$(m+1)\,Z[m+1] = \tfrac{1}{2}\,F_{uv}\big(\partial_u\,\partial_v\,Z[m] + \textstyle\sum_j(\partial_u\,Z[j])\cdot(\partial_v\,Z[m-j])\big).$$

Definition. A power series f in an auxiliary variable ϵ and in the z_i's, including $i\notin B$, is called *docile* if every monomial μ in it satisfies $\deg_z\mu \le 2\deg_\epsilon\mu + 2$; we will short that to $\deg_z f \le 2\deg_\epsilon f + 2$.

Claim 1. The synthesis equation preserves docility: if E is docile then so is Z_λ, and in particular, so is $\log\left\langle F:e^E\right\rangle$.

(And so it makes sense to restrict our attention to docile perturbations!)

Claim 2. Restricting attention to $\{z_i : i\in B\}$, if $\deg_{z_B}E \le 4\deg_\epsilon E$ then $\deg_{z_B}Z_\lambda \le 4\deg_\epsilon Z_\lambda - 2\deg_\lambda Z_\lambda$ and thus $\deg_\lambda Z_\lambda \le 2\deg_\epsilon Z_\lambda$.

Claim 2 implies that if $E\ |_{\epsilon=0}$ is independent of z_B and if we only care about Z_λ up to ϵ^k, then the iterative process for finding Z_λ terminates at $Z[2\,k]$.

Conclusion. We can compute efficiently (in polynomial time!) if all of our generating functions are of the form $\omega\,e^{Q+P}$, where ω is a scalar, Q is an ϵ-free quadratic, and $P = \sum_{k=1}^{\$k}P^{(k)}\,\epsilon^k$, where $\deg P^{(k)} \le 2\,k+2$.

On to the implementation...

$\mathbb{E}[\omega, Q, \text{P\_}\epsilon\text{Series}]$ represents $\omega\,e^{Q+P}$, where ω is a scalar, Q is an ϵ-free quadratic, and $P = \sum_{k=0}^{\$k}P[\![k]\!]\,\epsilon^k$ is a docile perturbation (it is ill-advised to include ω in P because then it will have log terms, so always, $P[\![0]\!] = 0$).

Initialization and minor utilities

(Alt) In[∘]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Classes\\21-1350-KnotTheory"];
Once[<< "Common.m"];
```

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
```

(Alt) In[∘]:=
```
$k=1;
```

(Alt) In[∘]:=
```
CCF[ℰ_] := ExpandDenominator@ExpandNumerator@Together[ℰ];
```

(Alt) In[∘]:= CCF$\left[\frac{1}{T-1}+\frac{1}{T-2}\right]$

(Alt) Out[∘]= $\frac{-3+2\,T}{2-3\,T+T^2}$

(Alt) In[∘]:= ComposeList$\left[\{\text{Together, ExpandNumerator, ExpandDenominator}\},\ \frac{1}{T-1}+\frac{1}{T-2}\right]$

(Alt) Out[∘]= $\left\{\frac{1}{-2+T}+\frac{1}{-1+T},\ \frac{-3+2\,T}{(-2+T)\times(-1+T)},\ \frac{-3+2\,T}{(-2+T)\times(-1+T)},\ \frac{-3+2\,T}{2-3\,T+T^2}\right\}$

(Alt) In[∘]:=
```
CF[ℰ_List] := CF /@ ℰ;
CF[ℰ_𝔢Series] := CF /@ ℰ;
CF[ℰ_] := Module[
    {vs = Cases[ℰ, (p | x | π | ξ)_, ∞] ⋃ {p | x | π | ξ}},
    Total[CoefficientRules[Expand[ℰ], vs] /. (ps_ → c_) :→ CCF[c] (Times @@ vs^ps)]
  ];
CF[ℰ_𝔼] := CF /@ ℰ;
CF[𝔼sp___[ℰs___]] := CF /@ 𝔼sp[ℰs];
```

(Alt) In[∘]:= CF$\left[(T\,x_1+(T-1)\,x_2)\left(\frac{p_1}{T-2}+p_2\right)\right]$

(Alt) Out[∘]= $\frac{T\,p_1\,x_1}{-2+T}+T\,p_2\,x_1+\frac{(-1+T)\,p_1\,x_2}{-2+T}+(-1+T)\,p_2\,x_2$

(Alt) In[∘]:=
```
ϵSeries /: S1_ϵSeries ≡ S2_ϵSeries :=
  Length[S1] == Length[S2] ∧ Inner[CF[#1] == CF[#2] &, S1, S2, And];
ϵSeries[0] := ϵSeries @@ Table[0, $k + 1];
ϵSeries /: S1_ϵSeries + S2_ϵSeries :=
  ϵSeries @@ Table[S1⟦k⟧ + S2⟦k⟧, {k, Min[Length@S1, Length@S2]}];
ϵSeries /: S1_ϵSeries * S2_ϵSeries := ϵSeries @@
   Table[Sum[S1⟦j + 1⟧ * S2⟦k - j + 1⟧, {j, 0, k}], {k, 0, Min[Length@S1, Length@S2] - 1}];
ϵSeries /: c_ * S_ϵSeries := (c #) & /@ S;
ϵSeries /: ∂_{vs___} S_ϵSeries := (s ↦ ∂_{vs} s) /@ S;
```

(Alt) In[∘]:=
```
ϵSeries[0, 2, 7, -1] ϵSeries[0, 3, 5, 19, 1350]
```

(Alt) Out[∘]=
```
ϵSeries[0, 0, 6, 31]
```

The Main Program

Variables and their duals:

(Alt) In[∘]:=
```
{p*, x*, π*, ξ*} = {π, ξ, p, x};
(vs_List)* := (v ↦ v*) /@ vs;
(u_i_)* := (u*)_i;
```

𝔼 operations:

(Alt) In[∘]:=
```
𝔼 /: 𝔼[ω1_, Q1_, P1_] ≡ 𝔼[ω2_, Q2_, P2_] := CF[ω1 == ω2] ∧ CF[Q1 == Q2] ∧ (P1 ≡ P2);
𝔼 /: 𝔼[ω1_, Q1_, P1_] 𝔼[ω2_, Q2_, P2_] := 𝔼[ω1 ω2, Q1 + Q2, P1 + P2];
𝔼_{d1_→r1_}[ℰ1s___] ≡ 𝔼_{d2_→r2_}[ℰ2s___] ^:= (d1 == d2) ∧ (r1 == r2) ∧ (𝔼[ℰ1s] ≡ 𝔼[ℰ2s]);
𝔼_{d1_→r1_}[ℰ1s___] 𝔼_{d2_→r2_}[ℰ2s___] ^:= 𝔼_{(d1⋃d2)→(r1⋃r2)} @@ (𝔼[ℰ1s] 𝔼[ℰ2s]);
```

Getting rid of the quadratic using lemma 1.

Lemma 1. $\left\langle F : \mathcal{E}\, e^{\frac{1}{2}\sum_{i,j\in B} G_{ij} z_i z_j}\right\rangle_B = \left\langle F(I-\mathrm{GF})^{-1} : \det(I-\mathrm{GF})^{-1/2}\,\mathcal{E}\right\rangle_B.$

Lemma 1. $\langle F : \mathcal{E}\rangle_B = \left\langle F(I-\mathrm{GF})^{-1} : \det(I-\mathrm{GF})^{-1/2}\,\mathcal{E} e^{-\frac{1}{2}\sum_{i,j\in B} G_{ij} z_i z_j}\right\rangle_B.$

(Alt) In[∘]:=
```
Zip1_{} = Identity;
Zip1_{vs_} @⟨ℱ_, 𝔼[ω_, Q_, P_]⟩ := Module[{I, F, G, u, v},
  I = IdentityMatrix@Length@vs;
  F = Table[∂_{u,v} ℱ, {u, vs*}, {v, vs*}];
  G = Table[∂_{u,v} Q, {u, vs}, {v, vs}];
  CF /@ ⟨
    vs*.F.Inverse[I - G.F].vs* / 2,
    𝔼[PowerExpand@Factor[ω Det[I - G.F]^(-1/2)], Q - vs.G.vs / 2, P]
   ⟩
 ]
```

(Alt) In[∘]:=
```
Ex1 = CF /@ ⟨2/2 ξ1^2, 𝔼[19, 1/2 {x1, x2}.( -3 2 ; 2 13 ).{x1, x2}, ϵSeries[0, x1]]⟩
```

(Alt) Out[∘]= $\left\langle \xi_1^2, \mathbb{E}\left[19, -\frac{3 x_1^2}{2} + 2 x_1 x_2 + \frac{13 x_2^2}{2}, \epsilon\text{Series}[0, x_1]\right]\right\rangle$

(Alt) In[∘]:=
```
Ex1 // Zip1_{x1}
```

(Alt) Out[∘]= $\left\langle \frac{\xi_1^2}{7}, \mathbb{E}\left[\frac{19}{\sqrt{7}}, 2 x_1 x_2 + \frac{13 x_2^2}{2}, \epsilon\text{Series}[0, x_1]\right]\right\rangle$

Getting rid of linear terms using Lemma 2.

Lemma 2. $\langle F:\mathcal{E}\, e^{\sum_{i\in B} y_i z_i}\rangle_B = e^{\frac{1}{2}\sum_{i,j\in B} F_{ij} z_i z_j} \left\langle F:\mathcal{E}\,|_{Z_B\to Z_B+F y_B}\right\rangle_B$.

(Alt) In[∘]:=
```
Zip2_{} = Identity;
Zip2_{vs_} @⟨ℱ_, 𝔼[ω_, Q_, P_]⟩ := Module[{F, Y, u, v},
  F = Table[∂_{u,v} ℱ, {u, vs*}, {v, vs*}];
  Y = Table[∂_v Q, {v, vs}];
  CF /@ ⟨ℱ, 𝔼[ω, Q - Y.vs + Y.F.Y / 2, P /. Thread[vs → vs + F.Y]]⟩
 ]
```

Dealing with Feynman diagrams without ever seeing them, using the synthesis equation and iteration. Write $Z_\lambda = \sum Z[m]\,\lambda^m$ and then $Z[0] = Z_0 = E$ and

$$Z[m+1] = \frac{1}{2\,(m+1)}\, F_{uv}\Big(\partial_u \partial_v Z[m] + \textstyle\sum_j (\partial_u Z[j]) \cdot (\partial_v Z[m-j])\Big),$$

and we only care to compute up to $Z[\![2\,\$k]\!]$:

(Alt) In[∘]:=
```
Zip3_{vs_} @⟨ℱ_, 𝔼[ω_, Q_, P_]⟩ := Module[{Z, u, v, m, j},
  Z[0] = P;
  For[m = 0, m < 2 $k, ++m,
   Z[m + 1] = CF[1/(2 (m + 1))
       Sum[∂_{u*,v*} ℱ (∂_{u,v} Z[m] + Sum[(∂_u Z[j]) (∂_v Z[m - j]), {j, 0, m}]), {u, vs}, {v, vs}]]
   ];
  𝔼[ω, Q, CF[Sum[Z[m], {m, 0, 2 $k}] /. Table[v → 0, {v, vs}]]]
 ]
```

(Alt) In[∘]:=
```
Block[{$k = 2}, ⟨1/2 ξ^2, 𝔼[1, 0, ϵSeries[0, 1/6 x^3, 0]]⟩ // Zip3_{x}]
```

(Alt) Out[∘]= $\mathbb{E}\left[1, 0, \epsilon\text{Series}\left[0, 0, \frac{5}{24}\right]\right]$

(Alt) In[∘]:= $\frac{1}{2^3} + \frac{1}{3!\,2}$

(Alt) Out[∘]= $\frac{5}{24}$

(Alt) In[∘]:= $\text{Block}\left[\{\$k = 4\},\ \left\langle \frac{1}{2}\,\xi^2,\ \mathbb{E}\left[1, 0, \epsilon\text{Series}\left[0, \frac{1}{6}\,x^3, 0, 0, 0\right]\right]\right\rangle \text{ // Zip3}_{\{x\}}\right]$

(Alt) Out[∘]= $\mathbb{E}\left[1, 0, \epsilon\text{Series}\left[0, 0, \frac{5}{24}, 0, \frac{5}{16}\right]\right]$

From “Cubic Multigraphs A005967” by Richard J. Mathar, https://oeis.org/A005967/a005967.pdf:

CUBIC MULTIGRAPHS A005967

RICHARD J. MATHAR

Abstract. These are illustrations of the undirected connected cubic (3-regular) multigraphs up to 10 vertices as counted in [1, A005967].

1. 2 vertices

1.1. 0 multi-edges 2 loops. (1 graphs)



1.2. 1 multi-edges 0 loops. (1 graphs)



Total: 2 graphs, 1 without loops.

2. 4 vertices

2.1. 0 multi-edges 0 loops. (1 graphs)



2.2. 0 multi-edges 3 loops. (1 graphs)



2.3. 1 multi-edges 1 loops. (1 graphs)



2.4. 1 multi-edges 2 loops. (1 graphs)



2.5. 2 multi-edges 0 loops. (1 graphs)



Total: 5 graphs, 2 without loops.

Date: November 12, 2018.
2010 *Mathematics Subject Classification.* Primary 05C30; Secondary 05C75, 81Q30.
Key words and phrases. Graph Enumeration, Combinatorics.

1

(Alt) In[∘]:= $\frac{1}{4!}+\frac{1}{3!\,2^3}+\frac{1}{2^3}+\frac{1}{2^4}+\frac{1}{2^4}$

(Alt) Out[∘]= $\frac{5}{16}$

(Alt) In[∘]:=
```
Block[{$k = 8}, ⟨1/2 ξ^2, 𝔼[1, 0, ϵSeries[0, 1/6 x^3, Sequence @@ Table[0, $k - 1]]]⟩ // Zip3_{x}]
```

(Alt) Out[∘]= $\mathbb{E}\left[1, 0, \epsilon\text{Series}\left[0, 0, \frac{5}{24}, 0, \frac{5}{16}, 0, \frac{1105}{1152}, 0, \frac{565}{128}\right]\right]$

Check out https://oeis.org/!

(Alt) In[∘]:=
```
Module[{l, m}, Log[Sum[l = 3 m / 2; ((3 m)! ϵ^m)/(2^l l! 6^m m!), {m, 0, 30, 2}] + O[ϵ]^31]]
```

(Alt) Out[∘]= $\frac{5\epsilon^2}{24}+\frac{5\epsilon^4}{16}+\frac{1105\epsilon^6}{1152}+\frac{565\epsilon^8}{128}+\frac{82825\epsilon^{10}}{3072}+\frac{19675\epsilon^{12}}{96}+\frac{1282031525\epsilon^{14}}{688128}+\frac{80727925\epsilon^{16}}{4096}+\frac{1683480621875\epsilon^{18}}{7077888}+\frac{13209845125\epsilon^{20}}{4096}+\frac{2239646759308375\epsilon^{22}}{46137344}+\frac{19739117098375\epsilon^{24}}{24576}+\frac{6320791709083309375\epsilon^{26}}{436207616}+\frac{32468078556378125\epsilon^{28}}{114688}+\frac{38362676768845045751875\epsilon^{30}}{6442450944}+O[\epsilon]^{31}$

(Alt) In[∘]:=
```
Zip_{vs_}[ℱ_, ℰ_] := ⟨ℱ, ℰ⟩ // Zip1_{vs} // Zip2_{vs} // Zip3_{vs}
```

(Alt) In[∘]:=
```
𝔼_{d1_→r1_}[ℰ1s___] // 𝔼_{d2_→r2_}[ℰ2s___] := Module[{is = r1 ⋂ d2, lvs},
  lvs = Flatten@Table[{x_{$@i}, p_{$@i}}, {i, is}];
  𝔼_{(d1⋃Complement[d2,is])→(r2⋃Complement[r1,is])} @@ (Zip_{lvs⋃lvs*}[lvs*.lvs, Times[
      𝔼[ℰ1s] /. Table[(v : x | p)_i → v_{$@i}, {i, is}],
      𝔼[ℰ2s] /. Table[(v : ξ | π)_i → v_{$@i}, {i, is}]
    ]])
 ]
```

The Basic Tensors

(Alt) In[∘]:=
```
η_{i_} := 𝔼_{{}→{i}}[1, 0, ϵSeries[0]];
m_{i_,j_→k_} := 𝔼_{{i,j}→{k}}[1, -ξ_i π_j + (π_i + π_j) p_k + (ξ_i + ξ_j) x_k, ϵSeries[0]]
```

(Alt) In[∘]:=
```
AllMonomials[{}, 0] = {1};
AllMonomials[{}, d_Integer] /; d > 0 := {};
AllMonomials[{v_, vs___}, d_Integer] :=
  Join @@ Table[v^(d-k) AllMonomials[{vs}, k], {k, 0, d}];
AllMonomials[vs_List, {d_}] := Join @@ Table[AllMonomials[vs, k], {k, 0, d}];
```

(Alt) In[∘]:=
```
AllMonomials[{x, y, z}, 2]
```
(Alt) Out[∘]=
$$\{x^2, x y, x z, y^2, y z, z^2\}$$

(Alt) In[∘]:=
```
Basis[js_List, m_] := Flatten@Outer[Times,
    AllMonomials[Table[p_j, {j, js}], m], AllMonomials[Table[x_j, {j, js}], m]];
Basis[js_List, {m_}] := Flatten@Table[Basis[js, k], {k, 0, m}]
```

(Alt) In[∘]:=
```
Basis[{i, j}, {2}]
```
(Alt) Out[∘]=
$$\{1, p_i x_i, p_i x_j, p_j x_i, p_j x_j, p_i^2 x_i^2, p_i^2 x_i x_j, p_i^2 x_j^2, p_i p_j x_i^2, p_i p_j x_i x_j, p_i p_j x_j^2, p_j^2 x_i^2, p_j^2 x_i x_j, p_j^2 x_j^2\}$$

(Alt) In[∘]:=
```
GenericCombination[bas_, c_] := bas.Table[c_j, {j, Length@bas}];
GenericCombination[bas_, c_k_] := bas.Table[c_{k,j}, {j, Length@bas}];
```

(Alt) In[∘]:=
```
GenericCombination[Basis[{i, j}, {2}], c_1]
```
(Alt) Out[∘]=
$$c_{1,1} + p_i x_i c_{1,2} + p_i x_j c_{1,3} + p_j x_i c_{1,4} + p_j x_j c_{1,5} + p_i^2 x_i^2 c_{1,6} + p_i^2 x_i x_j c_{1,7} + p_i^2 x_j^2 c_{1,8} + p_i p_j x_i^2 c_{1,9} + p_i p_j x_i x_j c_{1,10} + p_i p_j x_j^2 c_{1,11} + p_j^2 x_i^2 c_{1,12} + p_j^2 x_i x_j c_{1,13} + p_j^2 x_j^2 c_{1,14}$$

(Alt) In[∘]:=
```
R_{i_,j_} := 𝔼_{{}→{i,j}}[T^(1/2), (T - 1) (p_i - p_j) x_j , ϵSeries @@
    Prepend[0]@Table[GenericCombination[Basis[{i, j}, {k + 1}], c_k], {k, $k}]];
R̄_{i_,j_} := 𝔼_{{}→{i,j}}[T^(-1/2), (T^(-1) - 1) (p_i - p_j) x_j, ϵSeries @@
    Prepend[0]@Table[GenericCombination[Basis[{i, j}, {k + 1}], d_k], {k, $k}]];
C_{i_} := 𝔼_{{}→{i}}[T^(1/2), 0, ϵSeries @@ Prepend[0]@
     Table[GenericCombination[Basis[{i}, {k + 1}], e_k], {k, $k}]];
C̄_{i_} := 𝔼_{{}→{i}}[T^(-1/2), 0, ϵSeries @@ Prepend[0]@
     Table[GenericCombination[Basis[{i}, {k + 1}], f_k], {k, $k}]];
```

(Alt) In[∘]:=
```
{R_{1,2}, R̄_{1,2}, C_1, C̄_1}
```
(Alt) Out[∘]=
$$\Big\{\mathbb{E}_{\{\}\to\{1,2\}}\Big[\sqrt{T}, (-1+T)(p_1-p_2)x_2, \epsilon\text{Series}\big[0, c_{1,1} + p_1 x_1 c_{1,2} + p_1 x_2 c_{1,3} + p_2 x_1 c_{1,4} + p_2 x_2 c_{1,5} + p_1^2 x_1^2 c_{1,6} + p_1^2 x_1 x_2 c_{1,7} + p_1^2 x_2^2 c_{1,8} + p_1 p_2 x_1^2 c_{1,9} + p_1 p_2 x_1 x_2 c_{1,10} + p_1 p_2 x_2^2 c_{1,11} + p_2^2 x_1^2 c_{1,12} + p_2^2 x_1 x_2 c_{1,13} + p_2^2 x_2^2 c_{1,14}\big]\Big],$$
$$\mathbb{E}_{\{\}\to\{1,2\}}\Big[\frac{1}{\sqrt{T}}, \left(-1+\frac{1}{T}\right)(p_1-p_2)x_2, \epsilon\text{Series}\big[0, d_{1,1} + p_1 x_1 d_{1,2} + p_1 x_2 d_{1,3} + p_2 x_1 d_{1,4} + p_2 x_2 d_{1,5} + p_1^2 x_1^2 d_{1,6} + p_1^2 x_1 x_2 d_{1,7} + p_1^2 x_2^2 d_{1,8} + p_1 p_2 x_1^2 d_{1,9} + p_1 p_2 x_1 x_2 d_{1,10} + p_1 p_2 x_2^2 d_{1,11} + p_2^2 x_1^2 d_{1,12} + p_2^2 x_1 x_2 d_{1,13} + p_2^2 x_2^2 d_{1,14}\big]\Big],$$
$$\mathbb{E}_{\{\}\to\{1\}}\Big[\sqrt{T}, 0, \epsilon\text{Series}\big[0, e_{1,1} + p_1 x_1 e_{1,2} + p_1^2 x_1^2 e_{1,3}\big]\Big],$$
$$\mathbb{E}_{\{\}\to\{1\}}\Big[\frac{1}{\sqrt{T}}, 0, \epsilon\text{Series}\big[0, f_{1,1} + p_1 x_1 f_{1,2} + p_1^2 x_1^2 f_{1,3}\big]\Big]\Big\}$$

(Alt) In[∘]:=

$$
\begin{aligned}
&\mathbf{RMoves} := \{\\
&\quad (R_{1,2}\, R_{4,3}\, R_{5,6} \;/\!/\; m_{1,4\to 1} \;/\!/\; m_{2,5\to 2} \;/\!/\; m_{3,6\to 3}) \equiv (R_{2,3}\, R_{4,5}\, R_{1,6} \;/\!/\; m_{1,4\to 1} \;/\!/\; m_{2,5\to 2} \;/\!/\; m_{3,6\to 3}),\\
&\quad (R_{1,2}\, \overline{R}_{3,4} \;/\!/\; m_{1,3\to 1} \;/\!/\; m_{2,4\to 2}) \equiv (\eta_1\, \eta_2),\\
&\quad (C_1\, \overline{C}_2 \;/\!/\; m_{1,2\to 1}) \equiv \eta_1,\\
&\quad (R_{1,4}\, \overline{R}_{5,2}\, \overline{C}_3 \;/\!/\; m_{2,4\to 2} \;/\!/\; m_{1,3\to 1} \;/\!/\; m_{1,5\to 1}) \equiv \overline{C}_1\, \eta_2,\\
&\quad (C_3\, R_{1,2} \;/\!/\; m_{2,3\to 2} \;/\!/\; m_{2,1\to 1}) \equiv (\overline{C}_3\, R_{1,2} \;/\!/\; m_{1,3\to 1} \;/\!/\; m_{1,2\to 1}),\\
&\quad (\overline{C}_2\, R_{1,3} \;/\!/\; m_{1,2\to 1} \;/\!/\; m_{1,3\to 1}) \equiv \eta_1,\; (\overline{C}_2\, \overline{R}_{3,1} \;/\!/\; m_{1,2\to 1} \;/\!/\; m_{1,3\to 1}) \equiv \eta_1,\\
&\quad (C_2\, \overline{R}_{1,3} \;/\!/\; m_{1,2\to 1} \;/\!/\; m_{1,3\to 1}) \equiv \eta_1,\; (C_2\, R_{3,1} \;/\!/\; m_{1,2\to 1} \;/\!/\; m_{1,3\to 1}) \equiv \eta_1\\
&\}
\end{aligned}
$$

(Alt) In[∘]:= **RMoves**

(Alt) Out[∘]=

$$
\begin{aligned}
&\{3\, c_{1,1} + 2\, p_1\, x_1\, c_{1,2} + p_2\, x_1\, c_{1,4} + p_3\, x_1\, c_{1,4} + T\, p_3\, x_2\, c_{1,4} + p_1\, x_2\, (c_{1,2} - T\, c_{1,2} + c_{1,3} + c_{1,4} - T\, c_{1,4}) +\\
&\quad 2\, T\, p_3\, x_3\, c_{1,5} + p_2\, x_2\, (T\, c_{1,2} + c_{1,5}) + p_1\, x_3\, (T\, c_{1,2} - T^2\, c_{1,2} + 2\, c_{1,3} - T\, c_{1,3} + c_{1,5} - T\, c_{1,5}) +\\
&\quad p_2\, x_3\, (T\, c_{1,3} + T\, c_{1,4} - T^2\, c_{1,4} + c_{1,5} - T\, c_{1,5}) + 2\, p_1^2\, x_1^2\, c_{1,6} + p_1^2\, x_1\, x_2\, c_{1,7} +\\
&\quad p_1^2\, x_1\, x_3\, (2\, T\, c_{1,6} - 2\, T^2\, c_{1,6} + c_{1,7}) + p_1\, p_2\, x_1^2\, c_{1,9} + p_1\, p_3\, x_1^2\, c_{1,9} + T^2\, p_2\, p_3\, x_2^2\, c_{1,9} + p_1\, p_2\, x_1\, x_2\, c_{1,10} +\\
&\quad T\, p_1\, p_3\, x_1\, x_3\, c_{1,10} + T^2\, p_2\, p_3\, x_2\, x_3\, c_{1,10} + p_1\, p_2\, x_1\, x_3\, (2\, T\, c_{1,9} - 2\, T^2\, c_{1,9} + c_{1,10} - T\, c_{1,10}) +\\
&\quad T^2\, p_2\, p_3\, x_3^2\, c_{1,11} + p_1\, p_2\, x_2^2\, (2\, T\, c_{1,6} - 2\, T^2\, c_{1,6} + T\, c_{1,9} - T^2\, c_{1,9} + c_{1,11}) +\\
&\quad p_1\, p_2\, x_2\, x_3\, (2\, T\, c_{1,7} - 2\, T^2\, c_{1,7} + 2\, T\, c_{1,10} - 2\, T^2\, c_{1,10} + 2\, c_{1,11} - 2\, T\, c_{1,11}) + p_1\, p_2\, x_3^2\\
&\qquad (2\, T\, c_{1,8} - 2\, T^2\, c_{1,8} + T^2\, c_{1,9} - 2\, T^3\, c_{1,9} + T^4\, c_{1,9} + T\, c_{1,10} - 2\, T^2\, c_{1,10} + T^3\, c_{1,10} + c_{1,11} - T\, c_{1,11}) +\\
&\quad p_2^2\, x_1^2\, c_{1,12} + p_3^2\, x_1^2\, c_{1,12} + T^2\, p_3^2\, x_2^2\, c_{1,12} + p_1\, p_3\, x_2^2\, (T\, c_{1,9} - T^2\, c_{1,9} + 2\, T\, c_{1,12} - 2\, T^2\, c_{1,12}) +\\
&\quad p_1^2\, x_2^2\, (c_{1,6} - 2\, T\, c_{1,6} + T^2\, c_{1,6} + c_{1,8} + c_{1,9} - 2\, T\, c_{1,9} + T^2\, c_{1,9} + c_{1,12} - 2\, T\, c_{1,12} + T^2\, c_{1,12}) +\\
&\quad p_2^2\, x_1\, x_2\, c_{1,13} + T\, p_3^2\, x_1\, x_3\, c_{1,13} + T^2\, p_3^2\, x_2\, x_3\, c_{1,13} + p_2^2\, x_1\, x_3\, (2\, T\, c_{1,12} - 2\, T^2\, c_{1,12} + c_{1,13} - T\, c_{1,13}) +\\
&\quad p_1\, p_3\, x_2\, x_3\, (T\, c_{1,10} - T^2\, c_{1,10} + 2\, T\, c_{1,13} - 2\, T^2\, c_{1,13}) +\\
&\quad p_1^2\, x_2\, x_3\, (c_{1,7} - T\, c_{1,7} + 2\, c_{1,8} - 2\, T\, c_{1,8} + c_{1,10} - 2\, T\, c_{1,10} + T^2\, c_{1,10} + c_{1,13} - 2\, T\, c_{1,13} + T^2\, c_{1,13}) +\\
&\quad 2\, T^2\, p_3^2\, x_3^2\, c_{1,14} + p_2^2\, x_2^2\, (T^2\, c_{1,6} + c_{1,14}) + p_2^2\, x_2\, x_3\, (T^2\, c_{1,7} + T\, c_{1,13} - T^2\, c_{1,13} + 2\, c_{1,14} - 2\, T\, c_{1,14}) +\\
&\quad p_1\, p_3\, x_3^2\, (T\, c_{1,11} + 2\, T\, c_{1,14} - 2\, T^2\, c_{1,14}) + p_1^2\, x_3^2\, (T^2\, c_{1,6} - 2\, T^3\, c_{1,6} + T^4\, c_{1,6} + T\, c_{1,7} - 2\, T^2\, c_{1,7} + T^3\, c_{1,7} +\\
&\qquad\quad 2\, c_{1,8} - 4\, T\, c_{1,8} + 3\, T^2\, c_{1,8} + c_{1,11} - 2\, T\, c_{1,11} + T^2\, c_{1,11} + c_{1,14} - 2\, T\, c_{1,14} + T^2\, c_{1,14}) + p_2^2\, x_3^2\\
&\qquad (T^2\, c_{1,8} + T^2\, c_{1,12} - 2\, T^3\, c_{1,12} + T^4\, c_{1,12} + T\, c_{1,13} - 2\, T^2\, c_{1,13} + T^3\, c_{1,13} + c_{1,14} - 2\, T\, c_{1,14} + T^2\, c_{1,14}) ==\\
&3\, c_{1,1} + 2\, p_1\, x_1\, c_{1,2} + p_1\, x_3\, c_{1,3} + p_1\, x_2\, (c_{1,2} - T\, c_{1,2} + c_{1,3}) + T\, p_3\, x_1\, c_{1,4} + p_2\, x_1\, (2\, c_{1,4} - T\, c_{1,4}) +\\
&\quad p_3\, x_2\, (2\, T\, c_{1,4} - T^2\, c_{1,4}) + 2\, T\, p_3\, x_3\, c_{1,5} + p_2\, x_2\, (T\, c_{1,2} + c_{1,4} - 2\, T\, c_{1,4} + T^2\, c_{1,4} + c_{1,5}) +\\
&\quad p_2\, x_3\, (T\, c_{1,3} + c_{1,5} - T\, c_{1,5}) + 2\, p_1^2\, x_1^2\, c_{1,6} + p_1^2\, x_1\, x_3\, c_{1,7} + p_1^2\, x_1\, x_2\, (2\, c_{1,6} - 2\, T\, c_{1,6} + c_{1,7}) +\\
&\quad p_1^2\, x_2\, x_3\, (c_{1,7} - T\, c_{1,7}) + p_1^2\, x_3^2\, c_{1,8} + p_1^2\, x_2^2\, (c_{1,6} - 2\, T\, c_{1,6} + T^2\, c_{1,6} + c_{1,8}) + T\, p_1\, p_3\, x_1^2\, c_{1,9} +\\
&\quad p_1\, p_2\, x_1^2\, (2\, c_{1,9} - T\, c_{1,9}) + p_1\, p_3\, x_1\, x_2\, (2\, T\, c_{1,9} - 2\, T^2\, c_{1,9}) + p_1\, p_3\, x_2^2\, (T\, c_{1,9} - 2\, T^2\, c_{1,9} + T^3\, c_{1,9}) +\\
&\quad T\, p_1\, p_3\, x_1\, x_3\, c_{1,10} + p_1\, p_2\, x_1\, x_2\, (2\, c_{1,9} - 4\, T\, c_{1,9} + 2\, T^2\, c_{1,9} + c_{1,10}) + p_1\, p_2\, x_1\, x_3\, (c_{1,10} - T\, c_{1,10}) +\\
&\quad p_1\, p_3\, x_2\, x_3\, (T\, c_{1,10} - T^2\, c_{1,10}) + p_1\, p_2\, x_2\, x_3\, (c_{1,10} - 2\, T\, c_{1,10} + T^2\, c_{1,10}) + T\, p_1\, p_3\, x_3^2\, c_{1,11} +\\
&\quad p_1\, p_2\, x_2^2\, (c_{1,9} - 3\, T\, c_{1,9} + 3\, T^2\, c_{1,9} - T^3\, c_{1,9} + c_{1,11}) + p_1\, p_2\, x_3^2\, (c_{1,11} - T\, c_{1,11}) +\\
&\quad T^2\, p_3^2\, x_1^2\, c_{1,12} + p_2\, p_3\, x_1^2\, (2\, T\, c_{1,12} - 2\, T^2\, c_{1,12}) + p_2^2\, x_1^2\, (2\, c_{1,12} - 2\, T\, c_{1,12} + T^2\, c_{1,12}) +\\
&\quad p_3^2\, x_1\, x_2\, (2\, T^2\, c_{1,12} - 2\, T^3\, c_{1,12}) + p_2\, p_3\, x_1\, x_2\, (4\, T\, c_{1,12} - 8\, T^2\, c_{1,12} + 4\, T^3\, c_{1,12}) +\\
&\quad p_2\, p_3\, x_2^2\, (T^2\, c_{1,9} + 2\, T\, c_{1,12} - 6\, T^2\, c_{1,12} + 6\, T^3\, c_{1,12} - 2\, T^4\, c_{1,12}) +
\end{aligned}
$$

$$p_3^2\,x_2^2\,\left(2\,T^2\,c_{1,12}-2\,T^3\,c_{1,12}+T^4\,c_{1,12}\right)+T^2\,p_3^2\,x_1\,x_3\,c_{1,13}+$$
$$p_2^2\,x_1\,x_2\,\left(2\,c_{1,12}-6\,T\,c_{1,12}+6\,T^2\,c_{1,12}-2\,T^3\,c_{1,12}+c_{1,13}\right)+$$
$$p_2\,p_3\,x_1\,x_3\,\left(2\,T\,c_{1,13}-2\,T^2\,c_{1,13}\right)+p_2^2\,x_1\,x_3\,\left(c_{1,13}-2\,T\,c_{1,13}+T^2\,c_{1,13}\right)+$$
$$p_3^2\,x_2\,x_3\,\left(2\,T^2\,c_{1,13}-T^3\,c_{1,13}\right)+p_2^2\,x_2\,x_3\,\left(T^2\,c_{1,7}+c_{1,13}-3\,T\,c_{1,13}+3\,T^2\,c_{1,13}-T^3\,c_{1,13}\right)+$$
$$p_2\,p_3\,x_2\,x_3\,\left(T^2\,c_{1,10}+2\,T\,c_{1,13}-4\,T^2\,c_{1,13}+2\,T^3\,c_{1,13}\right)+2\,T^2\,p_3^2\,x_3^2\,c_{1,14}+$$
$$p_2^2\,x_2^2\,\left(T^2\,c_{1,6}+c_{1,12}-4\,T\,c_{1,12}+6\,T^2\,c_{1,12}-4\,T^3\,c_{1,12}+T^4\,c_{1,12}+c_{1,14}\right)+$$
$$p_2\,p_3\,x_3^2\,\left(T^2\,c_{1,11}+2\,T\,c_{1,14}-2\,T^2\,c_{1,14}\right)+p_2^2\,x_3^2\,\left(T^2\,c_{1,8}+c_{1,14}-2\,T\,c_{1,14}+T^2\,c_{1,14}\right),$$

$$c_{1,1}+d_{1,1}+p_1\,x_1\,\left(c_{1,2}+d_{1,2}+d_{1,4}-T\,d_{1,4}\right)+p_2\,x_1\,\left(c_{1,4}+T\,d_{1,4}\right)+$$
$$\frac{p_1\,x_2\,\left(-c_{1,2}+T\,c_{1,2}+c_{1,3}+T\,d_{1,3}+T\,d_{1,5}-T^2\,d_{1,5}\right)}{T}+\frac{p_2\,x_2\,\left(-c_{1,4}+T\,c_{1,4}+c_{1,5}+T^2\,d_{1,5}\right)}{T}+$$
$$p_1\,p_2\,x_1^2\,\left(c_{1,9}+T\,d_{1,9}+2\,T\,d_{1,12}-2\,T^2\,d_{1,12}\right)+p_2^2\,x_1^2\,\left(c_{1,12}+T^2\,d_{1,12}\right)+$$
$$p_1^2\,x_1^2\,\left(c_{1,6}+d_{1,6}+d_{1,9}-T\,d_{1,9}+d_{1,12}-2\,T\,d_{1,12}+T^2\,d_{1,12}\right)+$$
$$\frac{p_1\,p_2\,x_1\,x_2\,\left(-2\,c_{1,9}+2\,T\,c_{1,9}+c_{1,10}+T^2\,d_{1,10}+2\,T^2\,d_{1,13}-2\,T^3\,d_{1,13}\right)}{T}+$$
$$\frac{p_2^2\,x_1\,x_2\,\left(-2\,c_{1,12}+2\,T\,c_{1,12}+c_{1,13}+T^3\,d_{1,13}\right)}{T}+$$
$$\frac{p_1^2\,x_1\,x_2\,\left(-2\,c_{1,6}+2\,T\,c_{1,6}+c_{1,7}+T\,d_{1,7}+T\,d_{1,10}-T^2\,d_{1,10}+T\,d_{1,13}-2\,T^2\,d_{1,13}+T^3\,d_{1,13}\right)}{T}+$$
$$\frac{p_1\,p_2\,x_2^2\,\left(c_{1,9}-2\,T\,c_{1,9}+T^2\,c_{1,9}-c_{1,10}+T\,c_{1,10}+c_{1,11}+T^3\,d_{1,11}+2\,T^3\,d_{1,14}-2\,T^4\,d_{1,14}\right)}{T^2}+$$
$$\frac{p_2^2\,x_2^2\,\left(c_{1,12}-2\,T\,c_{1,12}+T^2\,c_{1,12}-c_{1,13}+T\,c_{1,13}+c_{1,14}+T^4\,d_{1,14}\right)}{T^2}+\frac{1}{T^2}\,p_1^2\,x_2^2\,\left(c_{1,6}-2\,T\,c_{1,6}+\right.$$
$$\left.T^2\,c_{1,6}-c_{1,7}+T\,c_{1,7}+c_{1,8}+T^2\,d_{1,8}+T^2\,d_{1,11}-T^3\,d_{1,11}+T^2\,d_{1,14}-2\,T^3\,d_{1,14}+T^4\,d_{1,14}\right)==0,$$

$$e_{1,1}+f_{1,1}+p_1\,x_1\,\left(e_{1,2}+f_{1,2}\right)+p_1^2\,x_1^2\,\left(e_{1,3}+f_{1,3}\right)==0,\;\frac{p_2\,x_1\,\left(T\,c_{1,4}+4\,c_{1,12}-4\,T\,c_{1,12}+T^2\,d_{1,4}\right)}{T^2}+$$
$$\frac{p_2\,x_2\,\left(-T\,c_{1,4}+T^2\,c_{1,4}+T\,c_{1,5}-4\,c_{1,12}+8\,T\,c_{1,12}-4\,T^2\,c_{1,12}+2\,c_{1,13}-2\,T\,c_{1,13}+T^3\,d_{1,5}\right)}{T^2}+$$
$$\frac{p_1\,p_2\,x_1^2\,\left(T\,c_{1,9}-2\,c_{1,12}+2\,T\,c_{1,12}+T^2\,d_{1,9}\right)}{T^2}+\frac{1}{T^2}\,p_1\,p_2\,x_1\,x_2$$
$$\left(-2\,T\,c_{1,9}+2\,T^2\,c_{1,9}+T\,c_{1,10}+4\,c_{1,12}-8\,T\,c_{1,12}+4\,T^2\,c_{1,12}-2\,c_{1,13}+2\,T\,c_{1,13}+T^3\,d_{1,10}\right)+$$
$$\frac{1}{T^2}\,p_1\,p_2\,x_2^2\,\left(T\,c_{1,9}-2\,T^2\,c_{1,9}+T^3\,c_{1,9}-T\,c_{1,10}+T^2\,c_{1,10}+T\,c_{1,11}-2\,c_{1,12}+6\,T\,c_{1,12}-\right.$$
$$\left.6\,T^2\,c_{1,12}+2\,T^3\,c_{1,12}+2\,c_{1,13}-4\,T\,c_{1,13}+2\,T^2\,c_{1,13}-2\,c_{1,14}+2\,T\,c_{1,14}+T^4\,d_{1,11}\right)+$$
$$\frac{p_2^2\,x_1^2\,\left(c_{1,12}+T^2\,d_{1,12}\right)}{T^2}+\frac{p_2^2\,x_1\,x_2\,\left(-2\,c_{1,12}+2\,T\,c_{1,12}+c_{1,13}+T^3\,d_{1,13}\right)}{T^2}+$$
$$\frac{p_2^2\,x_2^2\,\left(c_{1,12}-2\,T\,c_{1,12}+T^2\,c_{1,12}-c_{1,13}+T\,c_{1,13}+c_{1,14}+T^4\,d_{1,14}\right)}{T^2}+$$

$$\frac{T^2 c_{1,1} + T c_{1,4} - T^2 c_{1,4} + 2 c_{1,12} - 4 T c_{1,12} + 2 T^2 c_{1,12} + T^2 d_{1,1} + T^2 f_{1,1}}{T^2} +$$
$$\frac{p_1 x_1 \left(T^2 c_{1,2} - T c_{1,4} + T^2 c_{1,4} + 2 T c_{1,9} - 2 T^2 c_{1,9} - 4 c_{1,12} + 8 T c_{1,12} - 4 T^2 c_{1,12} + T^2 d_{1,2} + T^2 f_{1,2}\right)}{T^2} +$$
$$\frac{1}{T^2} p_1 x_2 \left(-T^2 c_{1,2} + T^3 c_{1,2} + T^2 c_{1,3} + T c_{1,4} - 2 T^2 c_{1,4} + T^3 c_{1,4} - T c_{1,5} + T^2 c_{1,5} - 2 T c_{1,9} + 4 T^2 c_{1,9} - 2 T^3 c_{1,9} + T c_{1,10} - T^2 c_{1,10} + 4 c_{1,12} - 12 T c_{1,12} + 12 T^2 c_{1,12} - 4 T^3 c_{1,12} - 2 c_{1,13} + 4 T c_{1,13} - 2 T^2 c_{1,13} + T^3 d_{1,3} - T^2 f_{1,2} + T^3 f_{1,2}\right) +$$
$$\frac{p_1^2 x_1^2 \left(T^2 c_{1,6} - T c_{1,9} + T^2 c_{1,9} + c_{1,12} - 2 T c_{1,12} + T^2 c_{1,12} + T^2 d_{1,6} + T^2 f_{1,3}\right)}{T^2} +$$
$$\frac{1}{T^2} p_1^2 x_1 x_2 \left(-2 T^2 c_{1,6} + 2 T^3 c_{1,6} + T^2 c_{1,7} + 2 T c_{1,9} - 4 T^2 c_{1,9} + 2 T^3 c_{1,9} - T c_{1,10} + T^2 c_{1,10} - 2 c_{1,12} + 6 T c_{1,12} - 6 T^2 c_{1,12} + 2 T^3 c_{1,12} + c_{1,13} - 2 T c_{1,13} + T^2 c_{1,13} + T^3 d_{1,7} - 2 T^2 f_{1,3} + 2 T^3 f_{1,3}\right) +$$
$$\frac{1}{T^2} p_1^2 x_2^2 \left(T^2 c_{1,6} - 2 T^3 c_{1,6} + T^4 c_{1,6} - T^2 c_{1,7} + T^3 c_{1,7} + T^2 c_{1,8} - T c_{1,9} + 3 T^2 c_{1,9} - 3 T^3 c_{1,9} + T^4 c_{1,9} + T c_{1,10} - 2 T^2 c_{1,10} + T^3 c_{1,10} - T c_{1,11} + T^2 c_{1,11} + c_{1,12} - 4 T c_{1,12} + 6 T^2 c_{1,12} - 4 T^3 c_{1,12} + T^4 c_{1,12} - c_{1,13} + 3 T c_{1,13} - 3 T^2 c_{1,13} + T^3 c_{1,13} + c_{1,14} - 2 T c_{1,14} + T^2 c_{1,14} + T^4 d_{1,8} + T^2 f_{1,3} - 2 T^3 f_{1,3} + T^4 f_{1,3}\right) == f_{1,1} + p_1 x_1 f_{1,2} + p_1^2 x_1^2 f_{1,3},$$

$$\frac{p_1^2 x_1^2 \left(T^2 c_{1,6} + T c_{1,7} + c_{1,8} + T^2 c_{1,9} + T c_{1,10} + c_{1,11} + T^2 c_{1,12} + T c_{1,13} + c_{1,14} + e_{1,3}\right)}{T^2} +$$
$$\frac{p_1 x_1 \left(T^2 c_{1,2} + T c_{1,3} + T^2 c_{1,4} + T c_{1,5} - 2 T c_{1,7} - 4 c_{1,8} - T c_{1,10} - 2 c_{1,11} + T e_{1,2} - 4 e_{1,3} + 4 T e_{1,3}\right)}{T^2} +$$
$$\frac{T^2 c_{1,1} - T c_{1,3} + 2 c_{1,8} + T^2 e_{1,1} - T e_{1,2} + T^2 e_{1,2} + 2 e_{1,3} - 4 T e_{1,3} + 2 T^2 e_{1,3}}{T^2} == c_{1,1} - c_{1,4} + 2 c_{1,12} + f_{1,1} + p_1 x_1 \left(T c_{1,2} + c_{1,3} + T c_{1,4} + c_{1,5} - 2 T c_{1,9} - c_{1,10} - 4 T c_{1,12} - 2 c_{1,13} + T f_{1,2}\right) + p_1^2 x_1^2 \left(T^2 c_{1,6} + T c_{1,7} + c_{1,8} + T^2 c_{1,9} + T c_{1,10} + c_{1,11} + T^2 c_{1,12} + T c_{1,13} + c_{1,14} + T^2 f_{1,3}\right),$$

$$c_{1,1} - c_{1,4} + 2 c_{1,12} + f_{1,1} + p_1 x_1 \left(T c_{1,2} + c_{1,3} + T c_{1,4} + c_{1,5} - 2 T c_{1,9} - c_{1,10} - 4 T c_{1,12} - 2 c_{1,13} + T f_{1,2}\right) + p_1^2 x_1^2 \left(T^2 c_{1,6} + T c_{1,7} + c_{1,8} + T^2 c_{1,9} + T c_{1,10} + c_{1,11} + T^2 c_{1,12} + T c_{1,13} + c_{1,14} + T^2 f_{1,3}\right) == 0,$$

$$d_{1,1} - T d_{1,3} + 2 T^2 d_{1,8} + f_{1,1} + f_{1,2} - T f_{1,2} + 2 f_{1,3} - 4 T f_{1,3} + 2 T^2 f_{1,3} + p_1 x_1 \left(d_{1,2} + T d_{1,3} + d_{1,4} + T d_{1,5} - 2 T d_{1,7} - 4 T^2 d_{1,8} - T d_{1,10} - 2 T^2 d_{1,11} + T f_{1,2} + 4 T f_{1,3} - 4 T^2 f_{1,3}\right) + p_1^2 x_1^2 \left(d_{1,6} + T d_{1,7} + T^2 d_{1,8} + d_{1,9} + T d_{1,10} + T^2 d_{1,11} + d_{1,12} + T d_{1,13} + T^2 d_{1,14} + T^2 f_{1,3}\right) == 0,$$

$$d_{1,1} - d_{1,4} + 2 d_{1,12} + e_{1,1} + \frac{p_1 x_1 \left(d_{1,2} + T d_{1,3} + d_{1,4} + T d_{1,5} - 2 d_{1,9} - T d_{1,10} - 4 d_{1,12} - 2 T d_{1,13} + e_{1,2}\right)}{T} + \frac{p_1^2 x_1^2 \left(d_{1,6} + T d_{1,7} + T^2 d_{1,8} + d_{1,9} + T d_{1,10} + T^2 d_{1,11} + d_{1,12} + T d_{1,13} + T^2 d_{1,14} + e_{1,3}\right)}{T^2} == 0,$$

$$\frac{p_1^2 x_1^2 \left(T^2 c_{1,6} + T c_{1,7} + c_{1,8} + T^2 c_{1,9} + T c_{1,10} + c_{1,11} + T^2 c_{1,12} + T c_{1,13} + c_{1,14} + e_{1,3}\right)}{T^2} +$$
$$\frac{p_1 x_1 \left(T^2 c_{1,2} + T c_{1,3} + T^2 c_{1,4} + T c_{1,5} - 2 T c_{1,7} - 4 c_{1,8} - T c_{1,10} - 2 c_{1,11} + T e_{1,2} - 4 e_{1,3} + 4 T e_{1,3}\right)}{T^2} +$$

$$\frac{T^2 c_{1,1} - T c_{1,3} + 2 c_{1,8} + T^2 e_{1,1} - T e_{1,2} + T^2 e_{1,2} + 2 e_{1,3} - 4 T e_{1,3} + 2 T^2 e_{1,3}}{T^2} == 0\}$$

Solving for *R*, *C*, $k = 1

(Alt) In[∘]:=
```
$k = 1;
{R_{1,2}, C_1}
unknowns = Cases[{R_{1,2}, R̄_{1,2}, C_1, C̄_1}, (c | d | e | f)_{$k,_}, ∞] // Union
```

(Alt) Out[∘]=
$$\{\mathbb{E}_{\{\}\to\{1,2\}}[\sqrt{T}, (-1+T)(p_1-p_2)x_2, \epsilon\text{Series}[0, c_{1,1} + p_1 x_1 c_{1,2} + p_1 x_2 c_{1,3} + p_2 x_1 c_{1,4} + p_2 x_2 c_{1,5} + p_1^2 x_1^2 c_{1,6} + p_1^2 x_1 x_2 c_{1,7} + p_1^2 x_2^2 c_{1,8} + p_1 p_2 x_1^2 c_{1,9} + p_1 p_2 x_1 x_2 c_{1,10} + p_1 p_2 x_2^2 c_{1,11} + p_2^2 x_1^2 c_{1,12} + p_2^2 x_1 x_2 c_{1,13} + p_2^2 x_2^2 c_{1,14}]], \mathbb{E}_{\{\}\to\{1\}}[\sqrt{T}, 0, \epsilon\text{Series}[0, e_{1,1} + p_1 x_1 e_{1,2} + p_1^2 x_1^2 e_{1,3}]]\}$$

(Alt) Out[∘]=
$$\{c_{1,1}, c_{1,2}, c_{1,3}, c_{1,4}, c_{1,5}, c_{1,6}, c_{1,7}, c_{1,8}, c_{1,9}, c_{1,10}, c_{1,11}, c_{1,12}, c_{1,13}, c_{1,14}, d_{1,1}, d_{1,2}, d_{1,3}, d_{1,4}, d_{1,5}, d_{1,6}, d_{1,7}, d_{1,8}, d_{1,9}, d_{1,10}, d_{1,11}, d_{1,12}, d_{1,13}, d_{1,14}, e_{1,1}, e_{1,2}, e_{1,3}, f_{1,1}, f_{1,2}, f_{1,3}\}$$

(Alt) In[∘]:=
```
Short[errors = CCF /@ Cases[RMoves, a_ == b_ :> a - b], 25]
```

(Alt) Out[∘]//Short=
$$\{T p_1 x_3 c_{1,2} - T^2 p_1 x_3 c_{1,2} + p_1 x_3 c_{1,3} - T p_1 x_3 c_{1,3} - p_2 x_1 c_{1,4} + T p_2 x_1 c_{1,4} + p_3 x_1 c_{1,4} - T p_3 x_1 c_{1,4} + p_1 x_2 c_{1,4} - T p_1 x_2 c_{1,4} - p_2 x_2 c_{1,4} + 2 T p_2 x_2 c_{1,4} - T^2 p_2 x_2 c_{1,4} - T p_3 x_2 c_{1,4} + T^2 p_3 x_2 c_{1,4} + T p_2 x_3 c_{1,4} - T^2 p_2 x_3 c_{1,4} + \ll 169 \gg + 4 T^2 p_2 p_3 x_2 x_3 c_{1,13} - 2 T^3 p_2 p_3 x_2 x_3 c_{1,13} - T^2 p_3^2 x_2 x_3 c_{1,13} + T^3 p_3^2 x_2 x_3 c_{1,13} + T p_2^2 x_3^2 c_{1,13} - 2 T^2 p_2^2 x_3^2 c_{1,13} + T^3 p_2^2 x_3^2 c_{1,13} + 2 p_2^2 x_2 x_3 c_{1,14} - 2 T p_2^2 x_2 x_3 c_{1,14} + p_1^2 x_3^2 c_{1,14} - 2 T p_1^2 x_3^2 c_{1,14} + T^2 p_1^2 x_3^2 c_{1,14} + 2 T p_1 p_3 x_3^2 c_{1,14} - 2 T^2 p_1 p_3 x_3^2 c_{1,14} - 2 T p_2 p_3 x_3^2 c_{1,14} + 2 T^2 p_2 p_3 x_3^2 c_{1,14}, \frac{\ll 1 \gg}{T^2}, \ll 1 \gg, \ll 3 \gg, \ll 1 \gg, \frac{\ll 1 \gg}{T^2}, \frac{\ll 1 \gg}{T^2}\}$$

(Alt) In[∘]:=
```
eqns = Thread[0 == Union @@ (CoefficientRules[#, {x_1, x_2, x_3, p_1, p_2, p_3}][[;; , 2]] & /@ errors)]
```

(Alt) Out[∘]=
$$\{0 == c_{1,4} - T c_{1,4}, 0 == -c_{1,4} + T c_{1,4}, 0 == T c_{1,4} - T^2 c_{1,4}, 0 == -c_{1,4} + 2 T c_{1,4} - T^2 c_{1,4},$$
$$0 == -T c_{1,4} + T^2 c_{1,4}, 0 == T c_{1,2} - T^2 c_{1,2} + c_{1,3} - T c_{1,3} + c_{1,5} - T c_{1,5},$$
$$0 == -2 c_{1,6} + 2 T c_{1,6}, 0 == 2 T c_{1,6} - 2 T^2 c_{1,6}, 0 == c_{1,9} - T c_{1,9},$$
$$0 == -c_{1,9} + T c_{1,9}, 0 == 2 T c_{1,9} - 2 T^2 c_{1,9}, 0 == -2 c_{1,9} + 4 T c_{1,9} - 2 T^2 c_{1,9},$$
$$0 == -2 T c_{1,9} + 2 T^2 c_{1,9}, 0 == 2 T c_{1,6} - 2 T^2 c_{1,6} - c_{1,9} + 4 T c_{1,9} - 4 T^2 c_{1,9} + T^3 c_{1,9},$$
$$0 == 2 T c_{1,8} - 2 T^2 c_{1,8} + T^2 c_{1,9} - 2 T^3 c_{1,9} + T^4 c_{1,9} + T c_{1,10} - 2 T^2 c_{1,10} + T^3 c_{1,10},$$
$$0 == 2 T c_{1,7} - 2 T^2 c_{1,7} - c_{1,10} + 4 T c_{1,10} - 3 T^2 c_{1,10} + 2 c_{1,11} - 2 T c_{1,11},$$
$$0 == T^2 c_{1,9} - T^3 c_{1,9} + 2 T c_{1,12} - 2 T^2 c_{1,12}, 0 == c_{1,12} - T^2 c_{1,12}, 0 == -c_{1,12} + 2 T c_{1,12} - T^2 c_{1,12},$$
$$0 == c_{1,9} - 2 T c_{1,9} + T^2 c_{1,9} + c_{1,12} - 2 T c_{1,12} + T^2 c_{1,12}, 0 == -2 T c_{1,12} + 2 T^2 c_{1,12},$$
$$0 == -4 T c_{1,12} + 8 T^2 c_{1,12} - 4 T^3 c_{1,12}, 0 == -2 c_{1,12} + 6 T c_{1,12} - 6 T^2 c_{1,12} + 2 T^3 c_{1,12},$$
$$0 == -2 T^2 c_{1,12} + 2 T^3 c_{1,12}, 0 == -T^2 c_{1,12} + 2 T^3 c_{1,12} - T^4 c_{1,12},$$
$$0 == -c_{1,12} + 4 T c_{1,12} - 6 T^2 c_{1,12} + 4 T^3 c_{1,12} - T^4 c_{1,12}, 0 == -2 T c_{1,12} + 6 T^2 c_{1,12} - 6 T^3 c_{1,12} + 2 T^4 c_{1,12},$$
$$0 == 2 T c_{1,13} - 2 T^2 c_{1,13}, 0 == T c_{1,13} - T^2 c_{1,13}, 0 == 2 T c_{1,12} - 2 T^2 c_{1,12} + T c_{1,13} - T^2 c_{1,13},$$
$$0 == 2 c_{1,8} - 2 T c_{1,8} + c_{1,10} - 2 T c_{1,10} + T^2 c_{1,10} + c_{1,13} - 2 T c_{1,13} + T^2 c_{1,13}, 0 == -2 T c_{1,13} + 2 T^2 c_{1,13},$$
$$0 == -2 T c_{1,13} + 4 T^2 c_{1,13} - 2 T^3 c_{1,13}, 0 == T^2 c_{1,12} - 2 T^3 c_{1,12} + T^4 c_{1,12} + T c_{1,13} - 2 T^2 c_{1,13} + T^3 c_{1,13},$$
$$0 == -T^2 c_{1,13} + T^3 c_{1,13}, 0 == -c_{1,13} + 4 T c_{1,13} - 4 T^2 c_{1,13} + T^3 c_{1,13} + 2 c_{1,14} - 2 T c_{1,14},$$
$$0 == 2 T c_{1,14} - 2 T^2 c_{1,14}, 0 == T^2 c_{1,6} - 2 T^3 c_{1,6} + T^4 c_{1,6} + T c_{1,7} - 2 T^2 c_{1,7} + T^3 c_{1,7} +$$

$c_{1,8} - 4\,T\,c_{1,8} + 3\,T^2\,c_{1,8} + c_{1,11} - 2\,T\,c_{1,11} + T^2\,c_{1,11} + c_{1,14} - 2\,T\,c_{1,14} + T^2\,c_{1,14},$

$0 == -2\,T\,c_{1,14} + 2\,T^2\,c_{1,14},\ 0 == c_{1,1} + d_{1,1},\ 0 == c_{1,1} - c_{1,4} + \frac{c_{1,4}}{T} + 2\,c_{1,12} + \frac{2\,c_{1,12}}{T^2} - \frac{4\,c_{1,12}}{T} + d_{1,1},$

$0 == c_{1,2} + c_{1,4} - \frac{c_{1,4}}{T} - 2\,c_{1,9} + \frac{2\,c_{1,9}}{T} - 4\,c_{1,12} - \frac{4\,c_{1,12}}{T^2} + \frac{8\,c_{1,12}}{T} + d_{1,2},$

$0 == \frac{c_{1,4}}{T} + \frac{4\,c_{1,12}}{T^2} - \frac{4\,c_{1,12}}{T} + d_{1,4},\ 0 == c_{1,2} + d_{1,2} + d_{1,4} - T\,d_{1,4},\ 0 == c_{1,4} + T\,d_{1,4},$

$0 == c_{1,2} - \frac{c_{1,2}}{T} + \frac{c_{1,3}}{T} + d_{1,3} + d_{1,5} - T\,d_{1,5},\ 0 == c_{1,4} - \frac{c_{1,4}}{T} + \frac{c_{1,5}}{T} + T\,d_{1,5},$

$0 == c_{1,4} - \frac{c_{1,4}}{T} + \frac{c_{1,5}}{T} - 4\,c_{1,12} - \frac{4\,c_{1,12}}{T^2} + \frac{8\,c_{1,12}}{T} + \frac{2\,c_{1,13}}{T^2} - \frac{2\,c_{1,13}}{T} + T\,d_{1,5},$

$0 == c_{1,6} + c_{1,9} - \frac{c_{1,9}}{T} + c_{1,12} + \frac{c_{1,12}}{T^2} - \frac{2\,c_{1,12}}{T} + d_{1,6},\ 0 == \frac{c_{1,9}}{T} - \frac{2\,c_{1,12}}{T^2} + \frac{2\,c_{1,12}}{T} + d_{1,9},$

$0 == 2\,c_{1,9} - \frac{2\,c_{1,9}}{T} + \frac{c_{1,10}}{T} + 4\,c_{1,12} + \frac{4\,c_{1,12}}{T^2} - \frac{8\,c_{1,12}}{T} - \frac{2\,c_{1,13}}{T^2} + \frac{2\,c_{1,13}}{T} + T\,d_{1,10},$

$0 == -2\,c_{1,9} + \frac{c_{1,9}}{T} + T\,c_{1,9} + c_{1,10} - \frac{c_{1,10}}{T} + \frac{c_{1,11}}{T} - 6\,c_{1,12} - \frac{2\,c_{1,12}}{T^2} +$

$\frac{6\,c_{1,12}}{T} + 2\,T\,c_{1,12} + 2\,c_{1,13} + \frac{2\,c_{1,13}}{T^2} - \frac{4\,c_{1,13}}{T} - \frac{2\,c_{1,14}}{T^2} + \frac{2\,c_{1,14}}{T} + T^2\,d_{1,11},$

$0 == \frac{c_{1,12}}{T^2} + d_{1,12},\ 0 == c_{1,9} + T\,d_{1,9} + 2\,T\,d_{1,12} - 2\,T^2\,d_{1,12},\ 0 == c_{1,12} + T^2\,d_{1,12},$

$0 == c_{1,6} + d_{1,6} + d_{1,9} - T\,d_{1,9} + d_{1,12} - 2\,T\,d_{1,12} + T^2\,d_{1,12},\ 0 == -\frac{2\,c_{1,12}}{T^2} + \frac{2\,c_{1,12}}{T} + \frac{c_{1,13}}{T^2} + T\,d_{1,13},$

$0 == 2\,c_{1,9} - \frac{2\,c_{1,9}}{T} + \frac{c_{1,10}}{T} + T\,d_{1,10} + 2\,T\,d_{1,13} - 2\,T^2\,d_{1,13},\ 0 == 2\,c_{1,12} - \frac{2\,c_{1,12}}{T} + \frac{c_{1,13}}{T} + T^2\,d_{1,13},$

$0 == 2\,c_{1,6} - \frac{2\,c_{1,6}}{T} + \frac{c_{1,7}}{T} + d_{1,7} + d_{1,10} - T\,d_{1,10} + d_{1,13} - 2\,T\,d_{1,13} + T^2\,d_{1,13},$

$0 == c_{1,9} + \frac{c_{1,9}}{T^2} - \frac{2\,c_{1,9}}{T} - \frac{c_{1,10}}{T^2} + \frac{c_{1,10}}{T} + \frac{c_{1,11}}{T^2} + T\,d_{1,11} + 2\,T\,d_{1,14} - 2\,T^2\,d_{1,14},$

$0 == c_{1,12} + \frac{c_{1,12}}{T^2} - \frac{2\,c_{1,12}}{T} - \frac{c_{1,13}}{T^2} + \frac{c_{1,13}}{T} + \frac{c_{1,14}}{T^2} + T^2\,d_{1,14},$

$0 == c_{1,6} + \frac{c_{1,6}}{T^2} - \frac{2\,c_{1,6}}{T} - \frac{c_{1,7}}{T^2} + \frac{c_{1,7}}{T} + \frac{c_{1,8}}{T^2} + d_{1,8} + d_{1,11} - T\,d_{1,11} + d_{1,14} - 2\,T\,d_{1,14} + T^2\,d_{1,14},$

$0 == d_{1,1} - d_{1,4} + 2\,d_{1,12} + e_{1,1},\ 0 == \frac{d_{1,2}}{T} + d_{1,3} + \frac{d_{1,4}}{T} + d_{1,5} - \frac{2\,d_{1,9}}{T} - d_{1,10} - \frac{4\,d_{1,12}}{T} - 2\,d_{1,13} + \frac{e_{1,2}}{T},$

$0 == c_{1,6} + \frac{c_{1,7}}{T} + \frac{c_{1,8}}{T^2} + c_{1,9} + \frac{c_{1,10}}{T} + \frac{c_{1,11}}{T^2} + c_{1,12} + \frac{c_{1,13}}{T} + \frac{c_{1,14}}{T^2} + \frac{e_{1,3}}{T^2},$

$0 == \frac{d_{1,6}}{T^2} + \frac{d_{1,7}}{T} + d_{1,8} + \frac{d_{1,9}}{T^2} + \frac{d_{1,10}}{T} + d_{1,11} + \frac{d_{1,12}}{T^2} + \frac{d_{1,13}}{T} + d_{1,14} + \frac{e_{1,3}}{T^2},$

$0 == c_{1,1} - \frac{c_{1,3}}{T} + \frac{2\,c_{1,8}}{T^2} + e_{1,1} + e_{1,2} - \frac{e_{1,2}}{T} + 2\,e_{1,3} + \frac{2\,e_{1,3}}{T^2} - \frac{4\,e_{1,3}}{T},$

$0 == c_{1,2} + \frac{c_{1,3}}{T} + c_{1,4} + \frac{c_{1,5}}{T} - \frac{2\,c_{1,7}}{T} - \frac{4\,c_{1,8}}{T^2} - \frac{c_{1,10}}{T} - \frac{2\,c_{1,11}}{T^2} + \frac{e_{1,2}}{T} - \frac{4\,e_{1,3}}{T^2} + \frac{4\,e_{1,3}}{T},$

$$0 == -\frac{c_{1,3}}{T} + c_{1,4} + \frac{2\,c_{1,8}}{T^2} - 2\,c_{1,12} + e_{1,1} + e_{1,2} - \frac{e_{1,2}}{T} + 2\,e_{1,3} + \frac{2\,e_{1,3}}{T^2} - \frac{4\,e_{1,3}}{T} - f_{1,1},$$

$$0 == c_{1,1} - c_{1,4} + 2\,c_{1,12} + f_{1,1},\ 0 == e_{1,1} + f_{1,1},\ 0 == e_{1,2} + f_{1,2},$$

$$0 == c_{1,2} - T\,c_{1,2} - c_{1,3} + \frac{c_{1,3}}{T} + c_{1,4} - T\,c_{1,4} - c_{1,5} + \frac{c_{1,5}}{T} - \frac{2\,c_{1,7}}{T} - \frac{4\,c_{1,8}}{T^2} + 2\,T\,c_{1,9} + c_{1,10} - \frac{c_{1,10}}{T} - \frac{2\,c_{1,11}}{T^2} + 4\,T\,c_{1,12} + 2\,c_{1,13} + \frac{e_{1,2}}{T} - \frac{4\,e_{1,3}}{T^2} + \frac{4\,e_{1,3}}{T} - T\,f_{1,2},$$

$$0 == T\,c_{1,2} + c_{1,3} + T\,c_{1,4} + c_{1,5} - 2\,T\,c_{1,9} - c_{1,10} - 4\,T\,c_{1,12} - 2\,c_{1,13} + T\,f_{1,2},$$

$$0 == -c_{1,2} + T\,c_{1,2} + c_{1,3} - 2\,c_{1,4} + \frac{c_{1,4}}{T} + T\,c_{1,4} + c_{1,5} - \frac{c_{1,5}}{T} + 4\,c_{1,9} - \frac{2\,c_{1,9}}{T} - 2\,T\,c_{1,9} - c_{1,10} + \frac{c_{1,10}}{T} + 12\,c_{1,12} + \frac{4\,c_{1,12}}{T^2} - \frac{12\,c_{1,12}}{T} - 4\,T\,c_{1,12} - 2\,c_{1,13} - \frac{2\,c_{1,13}}{T^2} + \frac{4\,c_{1,13}}{T} + T\,d_{1,3} - f_{1,2} + T\,f_{1,2},$$

$$0 == e_{1,3} + f_{1,3},\ 0 == -2\,c_{1,6} + 2\,T\,c_{1,6} + c_{1,7} - 4\,c_{1,9} + \frac{2\,c_{1,9}}{T} + 2\,T\,c_{1,9} + c_{1,10} - \frac{c_{1,10}}{T} - 6\,c_{1,12} - \frac{2\,c_{1,12}}{T^2} + \frac{6\,c_{1,12}}{T} + 2\,T\,c_{1,12} + c_{1,13} + \frac{c_{1,13}}{T^2} - \frac{2\,c_{1,13}}{T} + T\,d_{1,7} - 2\,f_{1,3} + 2\,T\,f_{1,3},$$

$$0 == d_{1,2} + T\,d_{1,3} + d_{1,4} + T\,d_{1,5} - 2\,T\,d_{1,7} - 4\,T^2\,d_{1,8} - T\,d_{1,10} - 2\,T^2\,d_{1,11} + T\,f_{1,2} + 4\,T\,f_{1,3} - 4\,T^2\,f_{1,3},$$

$$0 == c_{1,6} - T^2\,c_{1,6} + \frac{c_{1,7}}{T} - T\,c_{1,7} - c_{1,8} + \frac{c_{1,8}}{T^2} + c_{1,9} - T^2\,c_{1,9} + \frac{c_{1,10}}{T} - T\,c_{1,10} - c_{1,11} + \frac{c_{1,11}}{T^2} + c_{1,12} - T^2\,c_{1,12} + \frac{c_{1,13}}{T} - T\,c_{1,13} - c_{1,14} + \frac{c_{1,14}}{T^2} + \frac{e_{1,3}}{T^2} - T^2\,f_{1,3},$$

$$0 == T^2\,c_{1,6} + T\,c_{1,7} + c_{1,8} + T^2\,c_{1,9} + T\,c_{1,10} + c_{1,11} + T^2\,c_{1,12} + T\,c_{1,13} + c_{1,14} + T^2\,f_{1,3},$$

$$0 == d_{1,6} + T\,d_{1,7} + T^2\,d_{1,8} + d_{1,9} + T\,d_{1,10} + T^2\,d_{1,11} + d_{1,12} + T\,d_{1,13} + T^2\,d_{1,14} + T^2\,f_{1,3},$$

$$0 == c_{1,6} - 2\,T\,c_{1,6} + T^2\,c_{1,6} - c_{1,7} + T\,c_{1,7} + c_{1,8} + 3\,c_{1,9} - \frac{c_{1,9}}{T} - 3\,T\,c_{1,9} + T^2\,c_{1,9} - 2\,c_{1,10} + \frac{c_{1,10}}{T} + T\,c_{1,10} + c_{1,11} - \frac{c_{1,11}}{T} + 6\,c_{1,12} + \frac{c_{1,12}}{T^2} - \frac{4\,c_{1,12}}{T} - 4\,T\,c_{1,12} + T^2\,c_{1,12} - 3\,c_{1,13} - \frac{c_{1,13}}{T^2} + \frac{3\,c_{1,13}}{T} + T\,c_{1,13} + c_{1,14} + \frac{c_{1,14}}{T^2} - \frac{2\,c_{1,14}}{T} + T^2\,d_{1,8} + f_{1,3} - 2\,T\,f_{1,3} + T^2\,f_{1,3},$$

$$0 == d_{1,1} - T\,d_{1,3} + 2\,T^2\,d_{1,8} + f_{1,1} + f_{1,2} - T\,f_{1,2} + 2\,f_{1,3} - 4\,T\,f_{1,3} + 2\,T^2\,f_{1,3}\}$$

(Alt) In[∘]:= **{sol} = Solve[eqns, unknowns]**

Solve: Equations may not give solutions for all "solve" variables.

(Alt) Out[∘]=

$$\Big\{\Big\{c_{1,1} \to -\frac{c_{1,2}}{2} - \frac{c_{1,5}}{2\,T},\ c_{1,3} \to -T\,c_{1,2} - c_{1,5},\ c_{1,4} \to 0,\ c_{1,6} \to 0,\ c_{1,8} \to -\frac{1}{2} \times (1 - T)\,c_{1,10},\ c_{1,9} \to 0,$$

$$c_{1,11} \to -T\,c_{1,7} - \frac{1}{2} \times (-1 + 3\,T)\,c_{1,10},\ c_{1,12} \to 0,\ c_{1,13} \to 0,\ c_{1,14} \to 0,\ d_{1,1} \to \frac{c_{1,2}}{2} + \frac{c_{1,5}}{2\,T},$$

$$d_{1,2} \to -c_{1,2},\ d_{1,3} \to \frac{c_{1,2}}{T} + \frac{c_{1,5}}{T^2},\ d_{1,4} \to 0,\ d_{1,5} \to -\frac{c_{1,5}}{T^2},\ d_{1,6} \to 0,\ d_{1,7} \to -\frac{c_{1,7}}{T} - \frac{(-1 + T)\,c_{1,10}}{T^2},$$

$$d_{1,8} \to -\frac{(1 - T)\,c_{1,10}}{2\,T^3},\ d_{1,9} \to 0,\ d_{1,10} \to -\frac{c_{1,10}}{T^2},\ d_{1,11} \to \frac{c_{1,7}}{T^2} - \frac{(-1 - T)\,c_{1,10}}{2\,T^3},\ d_{1,12} \to 0,\ d_{1,13} \to 0,$$

$$d_{1,14} \to 0,\ e_{1,1} \to -\frac{c_{1,2}}{2} - \frac{c_{1,5}}{2\,T},\ e_{1,2} \to -\frac{c_{1,10}}{T},\ e_{1,3} \to 0,\ f_{1,1} \to \frac{c_{1,2}}{2} + \frac{c_{1,5}}{2\,T},\ f_{1,2} \to \frac{c_{1,10}}{T},\ f_{1,3} \to 0\Big\}\Big\}$$

(Alt) In[∘]:= **sol /. (a_ → b_) ⧴ (a = b)**

(Alt) Out[∘]=
$$\left\{-\frac{c_{1,2}}{2}-\frac{c_{1,5}}{2T},\ -T\,c_{1,2}-c_{1,5},\ 0,\ 0,\ -\frac{1}{2}\times(1-T)\,c_{1,10},\ 0,\ -T\,c_{1,7}-\frac{1}{2}\times(-1+3T)\,c_{1,10},\ 0,\ 0,\ 0,\ \frac{c_{1,2}}{2}+\frac{c_{1,5}}{2T},\ -c_{1,2},\ \frac{c_{1,2}}{T}+\frac{c_{1,5}}{T^2},\ 0,\ -\frac{c_{1,5}}{T^2},\ 0,\ -\frac{c_{1,7}}{T}-\frac{(-1+T)\,c_{1,10}}{T^2},\ -\frac{(1-T)\,c_{1,10}}{2T^3},\ 0,\ -\frac{c_{1,10}}{T^2},\ \frac{c_{1,7}}{T^2}-\frac{(-1-T)\,c_{1,10}}{2T^3},\ 0,\ 0,\ 0,\ -\frac{c_{1,2}}{2}-\frac{c_{1,5}}{2T},\ -\frac{c_{1,10}}{T},\ 0,\ \frac{c_{1,2}}{2}+\frac{c_{1,5}}{2T},\ \frac{c_{1,10}}{T},\ 0\right\}$$

(Alt) In[∘]:= $\{R_{1,2},\ \overline{R}_{1,2},\ C_1,\ \overline{C}_1\}$

(Alt) Out[∘]=
$$\Big\{\mathbb{E}_{\{\}\to\{1,2\}}\Big[\sqrt{T},\ (-1+T)\,(p_1-p_2)\,x_2,\ \epsilon\text{Series}\Big[0,\ -\frac{c_{1,2}}{2}+p_1x_1c_{1,2}+p_1x_2\,(-T\,c_{1,2}-c_{1,5})-\frac{c_{1,5}}{2T}+p_2x_2c_{1,5}+p_1^2x_1x_2c_{1,7}+p_1p_2x_1x_2c_{1,10}-\frac{1}{2}\times(1-T)\,p_1^2x_2^2c_{1,10}+p_1p_2x_2^2\left(-T\,c_{1,7}-\frac{1}{2}\times(-1+3T)\,c_{1,10}\right)\Big]\Big],$$
$$\mathbb{E}_{\{\}\to\{1,2\}}\Big[\frac{1}{\sqrt{T}},\ \left(-1+\frac{1}{T}\right)(p_1-p_2)\,x_2,\ \epsilon\text{Series}\Big[0,\ \frac{c_{1,2}}{2}-p_1x_1c_{1,2}+\frac{c_{1,5}}{2T}-\frac{p_2x_2c_{1,5}}{T^2}+p_1x_2\left(\frac{c_{1,2}}{T}+\frac{c_{1,5}}{T^2}\right)-\frac{p_1p_2x_1x_2c_{1,10}}{T^2}-\frac{(1-T)\,p_1^2x_2^2c_{1,10}}{2T^3}+p_1p_2x_2^2\left(\frac{c_{1,7}}{T^2}-\frac{(-1-T)\,c_{1,10}}{2T^3}\right)+p_1^2x_1x_2\left(-\frac{c_{1,7}}{T}-\frac{(-1+T)\,c_{1,10}}{T^2}\right)\Big]\Big],$$
$$\mathbb{E}_{\{\}\to\{1\}}\Big[\sqrt{T},\ 0,\ \epsilon\text{Series}\Big[0,\ -\frac{c_{1,2}}{2}-\frac{c_{1,5}}{2T}-\frac{p_1x_1c_{1,10}}{T}\Big]\Big],$$
$$\mathbb{E}_{\{\}\to\{1\}}\Big[\frac{1}{\sqrt{T}},\ 0,\ \epsilon\text{Series}\Big[0,\ \frac{c_{1,2}}{2}+\frac{c_{1,5}}{2T}+\frac{p_1x_1c_{1,10}}{T}\Big]\Big]\Big\}$$

(Alt) In[∘]:= **Cases[{R₁,₂, R̄₁,₂, C₁, C̄₁}, (c | d | e | f)$_{\$k,\_}$, ∞] // Union**

(Alt) Out[∘]= $\{c_{1,2},\ c_{1,5},\ c_{1,7},\ c_{1,10}\}$

(Alt) In[∘]:= $\{c_{1,2}=0,\ c_{1,5}=0,\ c_{1,7}=0,\ c_{1,10}=1\};$
$\{R_{1,2},\ \overline{R}_{1,2},\ C_1,\ \overline{C}_1\}$

(Alt) Out[∘]=
$$\Big\{\mathbb{E}_{\{\}\to\{1,2\}}\Big[\sqrt{T},\ (-1+T)\,(p_1-p_2)\,x_2,\ \epsilon\text{Series}\Big[0,\ p_1p_2x_1x_2+\frac{1}{2}\times(-1+T)\,p_1^2x_2^2+\frac{1}{2}\times(1-3T)\,p_1p_2x_2^2\Big]\Big],$$
$$\mathbb{E}_{\{\}\to\{1,2\}}\Big[\frac{1}{\sqrt{T}},\ \left(-1+\frac{1}{T}\right)(p_1-p_2)\,x_2,\ \epsilon\text{Series}\Big[0,\ -\frac{(-1+T)\,p_1^2x_1x_2}{T^2}-\frac{p_1p_2x_1x_2}{T^2}-\frac{(1-T)\,p_1^2x_2^2}{2T^3}-\frac{(-1-T)\,p_1p_2x_2^2}{2T^3}\Big]\Big],$$
$$\mathbb{E}_{\{\}\to\{1\}}\Big[\sqrt{T},\ 0,\ \epsilon\text{Series}\Big[0,\ -\frac{p_1x_1}{T}\Big]\Big],\ \mathbb{E}_{\{\}\to\{1\}}\Big[\frac{1}{\sqrt{T}},\ 0,\ \epsilon\text{Series}\Big[0,\ \frac{p_1x_1}{T}\Big]\Big]\Big\}$$

```
(Alt) In[∘]:= RMoves
(Alt) Out[∘]= {True, True, True, True, True, True, True, True, True}
```

Some Knot Theory at $k=1

```
(Alt) In[∘]:= ZF[Knot[3, 1]]
```

KnotTheory: Loading precomputed data in PD4Knots`.

(Alt) Out[∘]= $\mathbb{E}_{\{\}\to\{1\}}\left[\frac{T}{1-T+T^2}, 0, \epsilon\text{Series}\left[0, \frac{-2+3T-2T^2+T^3}{T-2T^2+3T^3-2T^4+T^5}\right]\right]$

```
(Alt) In[∘]:= ZF /@ {Knot[6, 1], Knot[9, 46]}
```

(Alt) Out[∘]= $\left\{\mathbb{E}_{\{\}\to\{1\}}\left[\frac{T}{2-5T+2T^2}, 0, \epsilon\text{Series}\left[0, \frac{-5+16T-10T^2-4T^3+3T^4}{4T-20T^2+33T^3-20T^4+4T^5}\right]\right], \mathbb{E}_{\{\}\to\{1\}}\left[\frac{T}{2-5T+2T^2}, 0, \epsilon\text{Series}\left[0, \frac{-7+28T-30T^2+8T^3+T^4}{4T-20T^2+33T^3-20T^4+4T^5}\right]\right]\right\}$

```
(Alt) In[∘]:= equiv = {Knot[10, 106], Knot[12, NonAlternating, 369]};
Length@Union@Echo[ZF /@ equiv]

(Alt) In[∘]:= equiv =
   {Knot[12, Alternating, 427], Knot[12, Alternating, 435], Knot[12, Alternating, 990]};
Length@Union[ZF /@ equiv]
```

Solving for *R*, *C*, $k = 2

```
In[∘]:= $k = 2;
{R1,2, C1}
Short[RMoves, 20]

In[∘]:= unknowns = Cases[{R1,2, R̄1,2, C1, C̄1}, (c | d | e | f)$k,_, ∞] // Union

In[∘]:= Short[errors = CCF /@ Cases[RMoves, a_ == b_ :> a - b], 25]

In[∘]:= Short[#, 10] &[eqns =
   Thread[0 == Union @@ (CoefficientRules[#, {x1, x2, x3, p1, p2, p3}][[ ;; , 2]] & /@ errors)]]

In[∘]:= {sol} = Solve[eqns, unknowns]

In[∘]:= sol /. (a_ → b_) :> (a = b)

In[∘]:= Cases[{R1,2, R̄1,2, C1, C̄1}, (c | d | e | f)$k,_, ∞] // Union

In[∘]:= {c2,2 = 0, c2,5 = 0, c2,7 = 0, c2,10 = 0, c2,16 = 0};
{R1,2, R̄1,2, C1, C̄1}

In[∘]:= RMoves
```

Some Knot Theory at $k=2

According to 12XingStats.nb at pensieve://Projects/SL2Invariant/k=2/ the following pair have equal ρ_1 but different ρ_2:

In[∘]:=
```
equiv = {Knot[10, 106], Knot[12, NonAlternating, 369]};
res21 = ZF /@ equiv
Simplify[res21⟦1⟧ ≡ res21⟦2⟧]
```

According to 12XingStats.nb at pensieve://Projects/SL2Invariant/k=2/ the following triple have equal ρ_1 but different ρ_2:

In[∘]:=
```
equiv =
  {Knot[12, Alternating, 427], Knot[12, Alternating, 435], Knot[12, Alternating, 990]};
Length@Union[res22 = ZF /@ equiv]
```

Solving for *R*, *C*, $k = 3

In[∘]:=
```
$k = 3;
Short[RMoves, 20]
```

In[∘]:=
```
unknowns = Cases[{R_{1,2}, R̄_{1,2}, C_1, C̄_1}, (c | d | e | f)_{$k,_}, ∞] // Union
```

In[∘]:=
```
Short[errors = CCF /@ Cases[RMoves, a_ == b_ ⧴ a - b], 25]
```

In[∘]:=
```
Short[#, 10] &[eqns =
   Thread[0 == Union @@ (CoefficientRules[#, {x_1, x_2, x_3, p_1, p_2, p_3}]⟦;; , 2⟧ & /@ errors)]]
```

In[∘]:=
```
{sol} = Solve[eqns, unknowns]
```

In[∘]:=
```
sol /. (a_ → b_) ⧴ (a = b)
```

In[∘]:=
```
Cases[{R_{1,2}, R̄_{1,2}, C_1, C̄_1}, (c | d | e | f)_{$k,_}, ∞] // Union
```

In[∘]:=
```
{c_{3,2} = 0, c_{3,5} = 0, c_{3,7} = 0, c_{3,10} = 0, c_{3,16} = 0, c_{3,32} = 1};
{R_{1,2}, R̄_{1,2}, C_1, C̄_1}
```

In[∘]:=
```
RMoves
```